

CURSO-ABERRATION

Sistema clásico Aberration desarrollado por Keith Fitschen. Es un sistema tendencial que inicia su posición al cerrar por encima de la Banda de Bollinger en el lado largo y al contrario para el lado corto.

INPUTS DE LA ESTRATEGIA:

- **Allow_Long (True):** Permite activar o desactivar la apertura de posiciones largas.
- **Allow_Short (True):** Permite activar o desactivar la apertura de posiciones cortas.
- **Length (35):** Número de barras usadas para calcular la Banda de Bollinger.
- **StdDevUp (2):** Controla el número de desviaciones de la banda alta de Bollinger.
- **StdDevDn (2):** Controla el número de desviaciones de la banda baja de Bollinger.
- **Start_Equity (100000):** Importe inicial de la cuenta. Usado para el MM.
- **MMVar_Start (100):** Multiplicador del capital inicial para el MM.
- **MMVar_Profits (100):** Multiplicador del beneficio del sistema para el MM.
- **Min_Size (1):** Cantidad mínima de contratos o acciones que el sistema puede abrir.
- **Max_Size (100000):** Cantidad máxima de contratos o acciones que el sistema puede abrir.
- **RoundTo (1):** Unidad a la que se redondea el MM.

VARIABLES:

- **UpBand (0):** Variable numérica que guarda el valor de la banda alta de Bollinger.
- **DnBand (0):** Variable numérica que almacena el valor de la banda baja de Bollinger.
- **Ave (0):** Variable numérica que almacena la media central.
- **Contratos (0):** Variable numérica que guarda las acciones o contratos que se abren.
- **Profits (0):** Variable numérica que almacena el beneficio acumulado.

MONEY MANAGEMENT:

1. Calculamos los beneficios acumulados (incluyendo posiciones abiertas) en la variable Profits.
2. Determinamos el número de contratos a operar, que se establece como un porcentaje de la cuenta inicial y otro porcentaje de los beneficios, partido por el riesgo que aquí se define como el valor nominal:

```
Contratos = ((Start_Equity * MMVar_Start * 0.01) + (Profits *  
MMVar_Profits * 0.01)) / Value1;
```

3. Hay que asegurar que el número de contratos se redondea y que esté dentro de los límites mínimos y máximos establecidos.

ESTRATEGIA DE ENTRADA

CÁLCULO DE LAS BANDAS DE BOLLINGER Y APLICACIÓN DE FILTROS:

- 1- Calcula el valor de la banda de Bollinger alta y la baja en UpBand y DnBand.

APERTURA DE POSICIONES:

1. LARGAS: Si Allow_Long es True entonces:
 - a. Si tenemos una posición diferente de larga y tenemos un cierre por encima de la Banda superior de Bollinger:
 - i. Lanzamos una orden a mercado para abrir posición larga en la siguiente barra.
2. CORTAS: Si Allow_Short es True entonces:
 - a. Si tenemos una posición diferente de corta y se produce un cierre por debajo de la Banda inferior de Bollinger:
 - i. Lanzamos una orden a mercado para abrir posición corta en la siguiente barra.

GESTIÓN DE POSICIONES ABIERTAS

SALIDA EN BANDA CENTRAL:

Tanto el lado largo como el corto tienen salida por stop en la media Ave.


SALIDA EN BANDA CONTRARIA:

Esta salida es redundante y únicamente se ejecutaría si diera error al lanzar la orden stop de salida en banda central. Daría orden de salida a mercado en la barra siguiente. Técnicamente no se ejecutaría nunca.

CÓDIGO EASYLANGUAGE (CREEMOS QUE SIRVE COMO PSEUDOCÓDIGO, PERO SI ALGUIEN NO COMPRENDE ALGO PUEDE PREGUNTARLO EN DISCORD)

```
Inputs:
    Allow_Long(True),
    Allow_Short(True),
    Length(35),
    StdDevUp(2),
    StdDevDn(2),

    Start_Equity(100000),
    MVar_Start(100),
```



```
MMVar_Profits (100),
Min_Size (1),
Max_Size (100000),
RoundTo (1);

Vars:      UpBand(0),
           DnBand(0),
           Ave(0),
           Contratos (0),
           Profits (0);

UpBand = BollingerBand(Close,Length,StdDevUp);
DnBand = BollingerBand(Close,Length,-StdDevDn);
Ave = Average(Close,Length);

{ Money Management }
Profits= NetProfit + OpenPositionProfit;

If AbsValue(Close * BigPointValue) > 0 Then
    Value1 = AbsValue(Close * BigPointValue)
Else
    Value1 = 0.01;

Contratos = ((Start_Equity * MMVar_Start * 0.01) + (Profits * MMVar_Profits
* 0.01)) / Value1;
Contratos = IntPortion(Contratos / RoundTo) * RoundTo;
Contratos = MaxList(Contratos, Min_Size);
Contratos = MinList(Contratos, Max_Size);

If Allow_Long then begin
    // Entra largo con un cierre por encima de la Banda Superior
    If MarketPosition <> 1 and Close > UpBand then Buy Contratos shares
next bar at market;
    If MarketPosition = 1 then begin
        // Salida por stop en la banda central
        Sell next bar at Ave Stop;
        If Close < DnBand then Sell next bar at market;
    End;
End;

If Allow_Short then begin
    // Entra corto con un cierre por debajo de la Banda Inferior
    If MarketPosition <> -1 and Close < DnBand then SellShort Contratos
shares next bar at market;
    If MarketPosition = -1 then begin
        // Salida por stop en la banda central
        BuytoCover next bar at Ave Stop;
        If Close > UpBand then BuytoCover next bar at market;
    End;
End;
```